
1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D21wk49 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_TC_6_1_1_1_EUTRA

	Reason for change
	Presently TC 6.1.1.1 (with IMS enabled) fails at Step 26Aa1 & Step29Aa1 , since optional IMS PDN is not handled 

	Summary of change
	Used function f_IP_Handling_StartPDN to handle IMS PDN on Cell1 & Cell13 at Step20 & Step28 respectively. 

	TTCN module
	LTE\6_1\Idle_PLMNSelection.ttcn

	MCC160 Comment
	



Before change:
function f_TC_6_1_1_1_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) CellPowerList_Type v_CellPowerList_AtT3;
    var float v_HpPlmnUpper;        //@sic R5-192714 sic@
    timer t_HpPlmn_lower;
    timer t_HpPlmn_upper;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell13, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)  //@sic R5s100675 cl. 4.2 Change 1 sic@
    };
    v_CellPowerList_AtT3 := { //@sic R5-140751 sic@
      cs_CellPower(eutra_Cell14, tsc_ServingCellRS_EPRE)
    };
    
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init(c1);//SIB2 & SIB3 are transmitted
    
    //Set Sys info and cell info for cell 12, 13 & 14 to be used @ creation
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell12, cs_HPLMN_001_11);  //PLMN1
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell13, cs_HPLMN_001_21);  //PLMN2
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell14, cs_HPLMN_001_31);  //PLMN3

    //Set TAC for Cell1, Cell12, Cell13 & Cell14
    //TAC for Cell1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC(eutra_Cell12, tsc_IdleMode_TAC_Cell12);
    f_EUTRA_CellInfo_SetTAC(eutra_Cell13, tsc_IdleMode_TAC_Cell13);
    f_EUTRA_CellInfo_SetTAC(eutra_Cell14, tsc_IdleMode_TAC_Cell14);

    //Configure and activate all cells
    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF
    f_EUTRA_CellConfig_Def(eutra_Cell12);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell13);
    f_EUTRA_CellConfig_Def(eutra_Cell14);  //As postamble will be executed on this cell

    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.1.3.1-2
    f_UT_USIM_Insert(UT, "36.523-1 Table 6.1.1.1.3.1-2");

    //Preamble: The UE is made to camp on Cell 12 and then in state Switched OFF (State 1)
    f_EUTRA_Preamble(eutra_Cell12, STATE2_IDLEUPDATE);
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell12, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@
    //Adjust cell configuration according to T1
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@
    //Power on UE
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12); // @sic R5s120065 sic@
    if (pc_ue_CategoryDL_M1){       //@sic R5-192714 sic@
        v_HpPlmnUpper := 7500.0;
    } else {
        v_HpPlmnUpper := 660.0;
    }
    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell12, nonProtocolTimer, 120.0));
    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMax(eutra_Cell12, nonProtocolTimer, v_HpPlmnUpper)) ; //@sic R5-115625 R5-140751 sic@
    t_IdleMode_GenericTimer.start;

    //@siclog "Step 3" siclog@
    //Receive RRCConnectionRequest on Cell 12
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell12);   // @sic R5-120721 change 3 sic@
    t_IdleMode_GenericTimer.stop;
    //* @verdict pass RRCConnectionRequest message received on Cell 1
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 3");

    //@siclog "Step 4 - 18" siclog@
    //Perform step 3 to 17 of registration procedure on cell 12
    f_EUTRA_IdleUpdated_WithoutRRCConnReq(eutra_Cell12, STATE2_IDLEUPDATE);

    //@siclog "Step 19" siclog@
    //Check if PLMN1 is indicated by UE
    f_UT_PLMN_Check(UT, "00111");
    //* @verdict pass PLMN1 indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 19");

    //@siclog "Step 20" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 1 after 120 seconds, but before 660 / 7500 (for Category M1) seconds (Note 2) from power on?
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionRequest))
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_lower time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
        }
      [] t_HpPlmn_lower.timeout
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell1, t_HpPlmn_upper)) {  // @sic R5-120721 change 3 sic@
            //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_lower time but before t_HpPlmn_upper
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
          }
          else {
            //* @verdict fail No RRCConnectionRequest received
            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
          }
        }
    }
    
    //@siclog "Step 21 - 25" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell1);
       
    // VOID @sic R5-140751 sic@
    //@sic "Step 26Aa1" sic@
    // The UE may attempt the generic test procedure for IMS Re-registration on EPS HPLMN as specified in TS 36.508 [18] subclause 4.5A.30.
    //@sic R5-201099 sic@
    f_EUTRA_WaitForIMSReRegister(eutra_Cell1);
    
    //@siclog "Step 26B" siclog@
    //Check if PLMN4 is indicated by UE
    //@sic R5-203812 sic@
    f_UT_PLMN_Check(UT, "00101");
    //* @verdict pass PLMN4 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 26B");

    //@siclog "Step 28" siclog@
    //Adjusts cell power level according to T2
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    
    //@siclog "Step 29" siclog@
    //Check if UE camps on Cell13
    f_EUTRA_IdleMode_TAU(eutra_Cell13);
    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell13, nonProtocolTimer, 120.0)); //@sic R5-115625 R5s190153 sic@
    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell12, nonProtocolTimer, v_HpPlmnUpper)); //@sic R5s190153 sic@

    //@sic "Step 29Aa1" sic@
    //The UE may attempt the generic test procedure for IMS Re-registration on EPS VPLMN as specified in TS 36.508 [18] subclause 4.5A.30.
    //@sic R5-201099 sic@
    f_EUTRA_WaitForIMSReRegister(eutra_Cell13);
    
    //@siclog "Step 30" siclog@
    //SS adjusts cell levels according to row T3 of table 6.1.1.1.3.2
    // @sic R5-140751 sic@
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 31" siclog@
    //Check if PLMN2 is indicated by UE
    f_UT_PLMN_Check(UT, "00121");
    //* @verdict pass PLMN2 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 31");
     
    //@siclog "Step 32" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 14 after 120 seconds, but before 660 / 7500 (for Category M1) seconds after step29 ? (Note 1 and 2)
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell14, cr_508_RRCConnectionRequest))
        {
          //* @verdict fail RRCConnectionRequest received within t_HpPlmn_upper time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 32");
        }
      [] t_HpPlmn_lower.timeout  //@sic R5-192714 sic@
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell14, t_HpPlmn_upper)) {  // @sic R5-120721 change 3 R5-140751 sic@
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 32");
          }
        }
    }
    
    //@siclog "Step 33 - 37" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell14);
    
    //@siclog "Step 38" siclog@
    //Check if PLMN3 is indicated by UE
    f_UT_PLMN_Check(UT, "00131");
    //* @verdict PLMN3 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 38");
       
    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE
    f_EUTRA_Postamble(eutra_Cell14, E1_IDLE);
  };

After change:

import from IP_PTC_CtrlMsgs all;

function f_TC_6_1_1_1_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) CellPowerList_Type v_CellPowerList_AtT3;
    var float v_HpPlmnUpper;        //@sic R5-192714 sic@
    timer t_HpPlmn_lower;
    timer t_HpPlmn_upper;
    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell13, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)  //@sic R5s100675 cl. 4.2 Change 1 sic@
    };
    v_CellPowerList_AtT3 := { //@sic R5-140751 sic@
      cs_CellPower(eutra_Cell14, tsc_ServingCellRS_EPRE)
    };
    
    //Initialise all cell, security and mobile parameters
    f_EUTRA_Init(c1);//SIB2 & SIB3 are transmitted
    
    //Set Sys info and cell info for cell 12, 13 & 14 to be used @ creation
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell12, cs_HPLMN_001_11);  //PLMN1
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell13, cs_HPLMN_001_21);  //PLMN2
    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell14, cs_HPLMN_001_31);  //PLMN3

    //Set TAC for Cell1, Cell12, Cell13 & Cell14
    //TAC for Cell1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC(eutra_Cell12, tsc_IdleMode_TAC_Cell12);
    f_EUTRA_CellInfo_SetTAC(eutra_Cell13, tsc_IdleMode_TAC_Cell13);
    f_EUTRA_CellInfo_SetTAC(eutra_Cell14, tsc_IdleMode_TAC_Cell14);

    //Configure and activate all cells
    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF
    f_EUTRA_CellConfig_Def(eutra_Cell12);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell13);
    f_EUTRA_CellConfig_Def(eutra_Cell14);  //As postamble will be executed on this cell

    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.1.3.1-2
    f_UT_USIM_Insert(UT, "36.523-1 Table 6.1.1.1.3.1-2");

    //Preamble: The UE is made to camp on Cell 12 and then in state Switched OFF (State 1)
    f_EUTRA_Preamble(eutra_Cell12, STATE2_IDLEUPDATE);
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell12, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@
    //Adjust cell configuration according to T1
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@
    //Power on UE
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell12); // @sic R5s120065 sic@
    if (pc_ue_CategoryDL_M1){       //@sic R5-192714 sic@
        v_HpPlmnUpper := 7500.0;
    } else {
        v_HpPlmnUpper := 660.0;
    }
    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell12, nonProtocolTimer, 120.0));
    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMax(eutra_Cell12, nonProtocolTimer, v_HpPlmnUpper)) ; //@sic R5-115625 R5-140751 sic@
    t_IdleMode_GenericTimer.start;

    //@siclog "Step 3" siclog@
    //Receive RRCConnectionRequest on Cell 12
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell12);   // @sic R5-120721 change 3 sic@
    t_IdleMode_GenericTimer.stop;
    //* @verdict pass RRCConnectionRequest message received on Cell 1
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 3");

    //@siclog "Step 4 - 18" siclog@
    //Perform step 3 to 17 of registration procedure on cell 12
    f_EUTRA_IdleUpdated_WithoutRRCConnReq(eutra_Cell12, STATE2_IDLEUPDATE);

    //@siclog "Step 19" siclog@
    //Check if PLMN1 is indicated by UE
    f_UT_PLMN_Check(UT, "00111");
    //* @verdict pass PLMN1 indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 19");

    //@siclog "Step 20" siclog@
	if(pc_IMS) {
	  f_IP_Handling_StartPDN(IP, PDN_1, cs_DrbInfo_EUTRA(eutra_Cell1, tsc_DRB1));
	}
    //Check: Does the UE send an RRCConnectionRequest on Cell 1 after 120 seconds, but before 660 / 7500 (for Category M1) seconds (Note 2) from power on?
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionRequest))
        {
          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_lower time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
        }
      [] t_HpPlmn_lower.timeout
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell1, t_HpPlmn_upper)) {  // @sic R5-120721 change 3 sic@
            //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_lower time but before t_HpPlmn_upper
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
          }
          else {
            //* @verdict fail No RRCConnectionRequest received
            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 20");
          }
        }
    }
    
    //@siclog "Step 21 - 25" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell1);
       
    // VOID @sic R5-140751 sic@
    //@sic "Step 26Aa1" sic@
    // The UE may attempt the generic test procedure for IMS Re-registration on EPS HPLMN as specified in TS 36.508 [18] subclause 4.5A.30.
    //@sic R5-201099 sic@
    f_EUTRA_WaitForIMSReRegistration(eutra_Cell1);
    
    //@siclog "Step 26B" siclog@
    //Check if PLMN4 is indicated by UE
    //@sic R5-203812 sic@
    f_UT_PLMN_Check(UT, "00101");
    //* @verdict pass PLMN4 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 26B");

    //@siclog "Step 28" siclog@
    //Adjusts cell power level according to T2
 f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    if(pc_IMS) {
	  f_IP_Handling_StartPDN(IP, PDN_1, cs_DrbInfo_EUTRA(eutra_Cell13, tsc_DRB1));
	}
    
    //@siclog "Step 29" siclog@
    //Check if UE camps on Cell13
    f_EUTRA_IdleMode_TAU(eutra_Cell13);
    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell13, nonProtocolTimer, 120.0)); //@sic R5-115625 R5s190153 sic@
    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell12, nonProtocolTimer, v_HpPlmnUpper)); //@sic R5s190153 sic@

    //@sic "Step 29Aa1" sic@
    //The UE may attempt the generic test procedure for IMS Re-registration on EPS VPLMN as specified in TS 36.508 [18] subclause 4.5A.30.
    //@sic R5-201099 sic@
    f_EUTRA_WaitForIMSReRegister(eutra_Cell13);
    
    //@siclog "Step 30" siclog@
    //SS adjusts cell levels according to row T3 of table 6.1.1.1.3.2
    // @sic R5-140751 sic@
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 31" siclog@
    //Check if PLMN2 is indicated by UE
    f_UT_PLMN_Check(UT, "00121");
    //* @verdict pass PLMN2 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 31");
     
    //@siclog "Step 32" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 14 after 120 seconds, but before 660 / 7500 (for Category M1) seconds after step29 ? (Note 1 and 2)
    alt {
      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell14, cr_508_RRCConnectionRequest))
        {
          //* @verdict fail RRCConnectionRequest received within t_HpPlmn_upper time
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 32");
        }
      [] t_HpPlmn_lower.timeout  //@sic R5-192714 sic@
        {
          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell14, t_HpPlmn_upper)) {  // @sic R5-120721 change 3 R5-140751 sic@
            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 32");
          }
        }
    }
    
    //@siclog "Step 33 - 37" siclog@
    //Perform TAU procedure steps 2 to 6
    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell14);
    
    //@siclog "Step 38" siclog@
    //Check if PLMN3 is indicated by UE
    f_UT_PLMN_Check(UT, "00131");
    //* @verdict PLMN3 is indicated by UE
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 6.1.1.1 Step 38");
       
    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE
    f_EUTRA_Postamble(eutra_Cell14, E1_IDLE);
  };
